
Abstract

In this note we present a definition for quarcles, roughly spoken in-
tersections and overlays of quads and circles. Furthermore we present a
related software tool as a part of our research in teaching programming.

Keywords: Teaching programming, Lean Software-Engineering, Struc-
tured Programming

1 Introduction

In this note we present a lean solution to a statement of task, intended to be
given to study groups in highschool and undergraduate courses (c.f. Börstler
& Sperber [1]):

A geometric figure in a two-dimensional plane
is a triangle,
a quad
or an overlay
or an intersection of two geometric figures.
Write a Program Geometric Figures !
Design a system with the possibility to create geometric figures and
to check if a given point is inside the area of a chosen geometric
figure.

It turned out, as seen so often, that OOP is not a panacea. Even nearly
graduate students struggle in designing simple software systems. This coin-
cides with the author’s observation over the last decade that more and more

simple projects in study groups of undergraduate courses fail. Often there are
already severe problems on designing simple user input and output. Partly
this is caused by state of the art languages together with large libraries for
colourful, clickable user-interaction. These seductive posibilities are very help-
ful for students to avoid solving the original task. The sense for pulling out
the essentials from irrelevant details is weakly shaped.

The ability to abstract decreases increasingly.

Teaching programming can be well done by presenting appropriately chosen
examples, Wirth [7]. In this sense, we are collecting material for our study
groups, and the work at hand presents a further item in our pool of examples
teaching the power of simplicity.

2 Quarcles

While studying literature on education in computer science, for the first time
the author met the concept described in the introduction, which in this note
is called quarcles. Roughly spoken, these objects are obtained by intersections
and overlays of quads and circles. Presentation at hand proposes a definition
of a quarcle and encourages related geometrical research. Furthermore we
describe a simple software tool1 for quarcle construction and representation.
The design of such a tool is part of our research in teaching programming.

Figure 1: cG-simulation of a quarcle.

1Computer Geometry system CircleGraphics cG

We briefly recall the notion of different norms for v = (x, y) ∈ R2. In
daily life, we are fine with the Euclidean Norm ‖v‖22 := x2 + y2. Alternatives
are the Manhattan-norm ‖v‖1 := |x| + |y| and the maximum-norm ‖v‖0 :=
max(|x|, |y|). In what follows we consider sets, so-called generalised circles,
given by

Bp,r
m := {v ∈ R2 | ‖v −m‖p ≤ r} , p ∈ {0, 1, 2} ,m = (mx,my) ∈ R2 , r > 0 .

This setting gives a unified view on the geometric objects relevant in this note.
A circle in a common sense is obtained by choosing the Euclidean-, a quad
by the maximum- and a rhombus by the Manhattan-norm. As will be seen
below, this allows for a simplified implementation of an is element decision in
our software-solution.

Announced as objects obtained by intersections and overlays of quads and
circles, we state the
Definition:

• A quarcle q0 of level 0 is a generalised circle Bp,r
m .

• A quarcle qnof level n is a set P1(q
n)@P2(q

n),

with @ ∈ {∩,∪, \} and Pi(q
n) quarcles of levels less than n denoting the parents

of qn.

Using this notation the task given to students by Börstler & Sperber [1],
compactly written, reads
Statement of Task: Design a Software tool deciding wether v ∈ qn.

“Is v Element of qn?” may be answered in pseudo code with obvious
meaning by the
Algorithm:

IF ‖v −m‖p ≤ r THEN IsIn:=TRUE; ELSE IsIn:=FALSE; RETURN IsIn

in the case n=0. For n>0 and @ ∈ {∩,∪, \} condition ‖v −m‖p ≤ r is simply
replaced by

v ∈ P1(q
n) ∧ v ∈ P2(q

n) , v ∈ P1(q
n) ∨ v ∈ P2(q

n) , v ∈ P1(q
n) ∧ v /∈ P2(q

n) .

Remark: Preliminaries, task and principal solution may be presented on one
page, followed by two pages presenting implementation including explanation.

3 Programming

The underlying programming philosophy in this work is influenced by Job’s
simplicity is the ultimate sophistication [2] and Wirth’s plea for lean software
[8], where he took Einstein’s Make it as simple as possible, but not simpler as a
mantra. Based on experiences since 1984 in Learning, Teaching [5] and Apply-
ing [4] programming languages - Pascal,C & C++, Java, Oberon - the choice
in this note is Oberon [3, 6]. Implementations employ the Oxford Oberon
Compiler2 written by M.Spivey.

In what follows, we present the design of a module providing the is element
decision mentioned above. The head starts with

MODULE Quarcles;

IMPORT (* modules for textual input *)

U:=Utilities,F:=Files,S:=Strings,C:=Conv;

CONST

Qmax = 64; (* max number of quarcles *)

Zmax = 80; (* max length of textzeile *)

inp = "quarcles.txt"; (* Name of quarcles code *)

As defined above, a quarcle of level 0 may be described by three REAL
and one INTEGER value, repesenting midpoint (x,y), radius r and norm p.
Otherwise, setting these values to 0, we need information how the parents
P1, P2 can be adressed. A CHAR may indicate, how they are combined, e.g.
+, ∗,− representing {∪,∩, \}. Consequently a quarcle in our implementation
is given by

TYPE Quarcle = POINTER TO QuarcleType;

QuarcleType = RECORD

x,y,r : REAL; p : INTEGER;

p1,p2 : Quarcle; op : CHAR

END;

Case distinction between level 0 or greater is done by r>0 or r=0. We
assume that data of nq quarcles are listed in an ARRAY after reading a textfile.

VAR nq : INTEGER; (* number of quarcles *)

ql : ARRAY Qmax OF Quarcle; (* QuarcleList *)

After these preparations the is element decision can be formulated by

PROCEDURE IsIn(q:Quarcle; x,y:REAL) : BOOLEAN;

VAR R : BOOLEAN;

BEGIN

IF q.r>0. THEN

R := IsInCircle(q,x,y)

ELSE

R := IsInMixed(q,x,y)

2www.oberon07.com, loaded 11.4.2020

END;

RETURN R

END IsIn;

where the procedures IsInCircle(q,x,y) and IsInMixed are implemented
as follows

PROCEDURE IsInCircle(q:Quarcle; x,y:REAL) : BOOLEAN;

VAR R : BOOLEAN; d,dx,dy,r : REAL;

BEGIN

dx := q.x-x; dx := ABS(dx); (* basic dist *)

dy := q.y-y; dy := ABS(dy); (* information *)

r := q.r;

IF (dx>dy) THEN d := dx ELSE d := dy; END; (* max norm *)

IF q.p = 1 THEN d := dx + dy; END; (* manhattan *)

IF q.p = 2 THEN d := dx*dx + dy*dy; r := r*r; END; (* euclidean *)

IF d<=r THEN R := TRUE; ELSE R := FALSE; END; RETURN R

END IsInCircle;

PROCEDURE IsInMixed (q:Quarcle; x,y:REAL) : BOOLEAN;

VAR R : BOOLEAN; p1,p2 : Quarcle;

BEGIN p1 := q.p1; p2 := q.p2; R := FALSE;

IF q.op = "+" THEN IF IsIn(p1,x,y) OR IsIn(p2,x,y) THEN R := TRUE; END;

ELSIF q.op = "*" THEN IF IsIn(p1,x,y) & IsIn(p2,x,y) THEN R := TRUE; END;

ELSIF q.op = "-" THEN IF IsIn(p1,x,y) & ~IsIn(p2,x,y) THEN R := TRUE; END;

END;

RETURN R

END IsInMixed;

Eventually a procedure is exported, enabling a client to decide wether a
given point is element of the ith quarcle in a given list.

PROCEDURE Check*(i : INTEGER; x,y : REAL) : BOOLEAN;

BEGIN DEC(i); RETURN IsIn(ql[i],x,y)

END Check;

The module closes with a procedure - details omitted here - reading and
converting the textual input file and its invocation in module’s body.

PROCEDURE Init(); (* quarcle description from file to program *)

BEGIN Init;

END Quarcles.

An example of standard input file quarcles.txt is
440. 280. 200. 1 Diamond

300. 25. 200. 0 Quad

1 * 2

1 + 2

0. 200. 200. 2 Circle

4 + 5

6 - 2

organised as described above.

4 Graphical simulation

Now, we present the implementation of our graphic tool cG, generating a
virtual window. In more detail, each point in an rectangular area is checked to
be element of a given quarcle. Result is a textfile containing 0 or 1 information
for each point. This format can easily be visualised e.g. with gnuplot.

MODULE CircleGraphics;

IMPORT Args,Conv,F:=Files,Q:=Quarcles;

CONST

Width = 640; Height = 480; (* for graphical output *)

out = "cG.dat"; (* for data output *)

VAR

VW : ARRAY Height+1 OF ARRAY Width+1 OF BOOLEAN; (* Virtual Window *)

x,y : REAL; i,j,v,q : INTEGER;

s : ARRAY 20 OF CHAR; fp : F.File;

BEGIN q := 1;

IF Args.argc>=2 THEN Args.GetArg(1,s); q := Conv.IntVal(s); END;

x := 0.; y := 0.;

FOR i:=0 TO Width DO x := FLT(i);

FOR j:=0 TO Height DO y := FLT(j); VW[j][i] := Q.Check(q,x,y); END;

END;

fp := F.Open(out,"w");

FOR i:=0 TO Height DO

FOR j:=0 TO Width DO

IF VW[i][j] THEN v:=0; ELSE v:=1; END; F.WriteInt(fp,v,2);

END;

F.WriteLn(fp);

END;

F.Close(fp);

END CircleGraphics.

Using the input file quarcles.txt, quarcle 7 is shown in Figure 1.

